
Széchenyi István University  Department of  

 GIVK  Applied Mechanics 

 

Dynamics of Machines Week 10 – Exercise 

 
Vibration Analysis of 3D Model of a Machine and Machine Base  

 
 

Equation of motion:  2

i i i i i i aia 2 a a Q i 1,2,3,4,5,6       

Rearrangement:  2

i i i i i i aia 2 a a Q i 1,2,3,4,5,6        

Mass of the machine and machine base is m 500kg , non zero elements of the inertia tensor 
2

SxJ 1900kgm , 

2

SyJ 1700kgm , 
2

SzJ 1500kgm , spring constant values are the same at every vertices 
5

xic 5 10 m / N  , 

5

yic 5 10 m / N  , 
5

zic 2 10 m / N  , ( i 1,2,3,4 ).  

 

Coordinates of the vertices:  

(
1x 0,6m , 

1y 0,7m  , 
1z 0,4m  ), (

2x 0,6m , 
2y 0,8m , 

2z 0,4m  ), 

(
3x 0,5m  , 

3y 0,8m , 
3z 0,4m  ), (

4x 0,5m  , 
4y 0,7m  , 

4z 0,4m  ). 

 

Excitation angular frequency: 62,8 ( rad / s )    

Coordinates of excitation force: 
zF 500cos(62,8t )  

Coordinates of the position vector of excitation force, 
Fr : 

Fx 0,3m  , 
Fy 0,2m , 

Fz 0,4m  

 

Lehr damping coefficient in every natural frequency: 0,1    

 

Tasks: Determine the displacement of point P as a function of time!  

 Position vector of point P:   mPr (0,5 i 0,6 j 0,4 k )    

  Final time 
maxt 15 ( s ) . The machine revs up from rest to the working engine speed during 

startt 3( s ) . 

Length of a time step is t 0,001 ( s )  . 

Solving method: The problem will be solved in two steps. In the first step the undamped natural frequencies (eigenvalues) 

and eigenvectors will be determined and in the second step the uncoupled (independent) damped differential equations will 

be solved with Runge-Kutta method.  

 

 
 

// week 10 - exercise  

clear; 

usecanvas(%F); 

//stacksize('max')   

 

// Variables:  

// mass (kg) and moment of inertia variables (kgm^2) 

m=500; Jsx=1900; Jsy=1700; Jsz=1500; 

// spring constants (m/N) 

cxi=5.0e-5; cyi=5.0e-5; czi=2.0e-5; 

// coordinates of the 4 vertices (m) 

x1= 0.6; y1=-0.7; z1=-0.4;  

x2= 0.6; y2= 0.8; z2=-0.4; 

x3=-0.5; y3= 0.8; z3=-0.4; 

x4=-0.5; y4=-0.7; z4=-0.4; 

// Coordinates of the position vector of excitation force (m) 

xF=-0.3; yF= 0.2; zF= 0.4; 

// Coordinates of the position vector of point P  

xP=+0.5; yP= 0.6; zP= 0.4; 

// Lehr damping in every natural frequency 

xsi=zeros(6,1); 

xsi(1,1)=0.1; 

xsi(2,1)=0.1; 

xsi(3,1)=0.1; 

xsi(4,1)=0.1; 

xsi(5,1)=0.1; 

xsi(6,1)=0.1; 

 

F=zeros(3,1); F(3,1)= 500;   // coordinates of the force vector (N) 

omega=62.8;   // excitation angular frequency (rad/s)  

 

// Initial conditions matrix (initial displacement and initial velocity values) 

abi0(1,1)=0;abi0(2,1)=0;abi0(3,1)=0;abi0(4,1)=0;abi0(5,1)=0;abi0(6,1)=0; 

abi0(7,1)=0;abi0(8,1)=0;abi0(9,1)=0;abi0(10,1)=0;abi0(11,1)=0;abi0(12,1)=0; 

// time variables:  

tmax=15; // final time (s) 

tstart=3;  // (s) 

dt=0.001; // time step (s)  

t0=0;  // initial time (s) 

t=0:dt:tmax; // time interval (s) 

// multiplier matrices of position vectors   

R1=zeros(3,3);  

R1(1,2)= z1;  R1(1,3)=-y1;  

R1(2,1)=-z1;  R1(2,3)= x1;  

R1(3,1)= y1;  R1(3,2)=-x1;  

R2=zeros(3,3);  

R2(1,2)= z2;  R2(1,3)=-y2;  

R2(2,1)=-z2;  R2(2,3)= x2;  

R2(3,1)= y2;  R2(3,2)=-x2;  

R3=zeros(3,3);  

R3(1,2)= z3;  R3(1,3)=-y3;  

R3(2,1)=-z3;  R3(2,3)= x3;  

R3(3,1)= y3;  R3(3,2)=-x3;  

R4=zeros(3,3);  

R4(1,2)= z4;  R4(1,3)=-y4;  

R4(2,1)=-z4;  R4(2,3)= x4;  

R4(3,1)= y4;  R4(3,2)=-x4; 

 

RF=zeros(3,3);  

RF(1,2)= zF;  RF(1,3)=-yF;  

RF(2,1)=-zF;  RF(2,3)= xF;  

RF(3,1)= yF;  RF(3,2)=-xF; 

 

 

RP=zeros(3,3);  

RP(1,2)= zP;  RP(1,3)=-yP;  

RP(2,1)=-zP;  RP(2,3)= xP;  

RP(3,1)= yP;  RP(3,2)=-xP; 

 

// mass matrix 

M=zeros(6,6); 

M(1,1)=m;    M(2,2)=m;    M(3,3)=m; 

M(4,4)=Jsx;  M(5,5)=Jsy;  M(6,6)=Jsz; 

MM=M; 

 

// spring matrix  

C=zeros(6,6); 

Ci=zeros(3,3); 

Ci(1,1)=1/cxi;    Ci(2,2)=1/cyi;    Ci(3,3)=1/czi; 

/////// stiffness matrix  

C(1:3,1:3)=Ci+Ci+Ci+Ci; 

C(1:3,4:6)=Ci*R1+Ci*R2+Ci*R3+Ci*R4; 

C(4:6,1:3)=R1'*Ci+R2'*Ci+R3'*Ci+R4'*Ci; 

C(4:6,4:6)=R1'*Ci*R1+R2'*Ci*R2+R3'*Ci*R3+R4'*Ci*R4; 

CC=C; 

 

// generalized eigenvalue problem 

[al,be,X]=spec(C,M); 

 

alfa=zeros(6,1); 

alfa2=zeros(6,1); 

 

for i=1:6 

// eigenvectors normalization 

di2=X(:,i)'*MM*X(:,i);  

di=sqrt(di2); 

X(:,i)=X(:,i)/di; 

// natural frequency squares 

alfa2(i,1)=X(:,i)'*CC*X(:,i); 

// natural frequencies 

alfa(i,1)=sqrt(alfa2(i,1)); 

end   

alfa_diffmethod=sqrt(al./be);    //natural frequencies with a different method 

 

// real part of complex numbers 

X=real(X); 

alfa2=real(alfa2); 

alfa=real(alfa) 

 

// generalized force  

QF0=zeros(6,1); 

QF0(1:3,1)=F; 

QF0(4:6,1)=RF'*F; 

 

// generalized force in the modal space  

QF1=X'*QF0; 

 

function dabi=equation(t, abi) 

fact=1; 

if t < tstart then 

 fact=t/tstart; 

end   

dabi(1)=abi(2); 

dabi(2)=-2*xsi(1,1)*alfa(1,1)*abi(2)-alfa2(1,1)*abi(1)+QF1(1,1)*cos(fact*omega*t); 

dabi(3)=abi(4); 

dabi(4)=-2*xsi(2,1)*alfa(2,1)*abi(4)-alfa2(2,1)*abi(3)+QF1(2,1)*cos(fact*omega*t); 

dabi(5)=abi(6); 

dabi(6)=-2*xsi(3,1)*alfa(3,1)*abi(6)-alfa2(3,1)*abi(5)+QF1(3,1)*cos(fact*omega*t); 

dabi(7)=abi(8); 

dabi(8)=-2*xsi(4,1)*alfa(4,1)*abi(8)-alfa2(4,1)*abi(7)+QF1(4,1)*cos(fact*omega*t); 

dabi(9)=abi(10); 

dabi(10)=-2*xsi(5,1)*alfa(5,1)*abi(10)-alfa2(5,1)*abi(9)+QF1(5,1)*cos(fact*omega*t); 

dabi(11)=abi(12); 

dabi(12)=-2*xsi(6,1)*alfa(6,1)*abi(12)-alfa2(6,1)*abi(11)+QF1(6,1)*cos(fact*omega*t); 

endfunction 

 

// differential equation solving  

abi=ode(abi0,t0,t,equation); 

 

// sorting displacement and velocity  

for i=1:6 

ai(i,:)=abi(2*i-1,:); // normalized displacement  

bi(i,:)=abi(2*i,:); // normalized velocity 

end 

 

// general displacement coordinates  

qi=X*ai; 

uvw=qi(1:3,:); 

fixyz=qi(4:6,:); 

 

// displacement of point P 

uvwP=uvw+RP*fixyz; 

 

subplot(2,1,1); 

plot(t,uvwP(3,:)*1000) 

xtitle("Displacement function of point P","time (s)","displacement (mm)") 

xgrid(2) 

 

 

qdi=X*bi; 

duvw=qdi(1:3,:); 

dfixyz=qdi(4:6,:);  

 

// velocity of point P 

duvwP=duvw+RP*dfixyz; 

 

subplot(2,1,2); 

plot(t,duvwP(3,:)) 

xtitle("Velocity function of point P ","time (s)","velocity (m/s)") 

xgrid(2) 

